TM No. 861032

NAVAL UNDERWATER SYSTEMS CENTER
NEW LONDON LABORATORY
NEW LONDON, CONNECTICUT 06320

Technical Memorandum

COMPUTATION OF COMPLEX AIRY FUNCTIONS

Date: 19 March 1986

Prepared by: Marvin J. Goldstein
Marvin J. Goldstein
Surface Ship Sonar Dept

Report Documentation Page

Form Approved
OMB No. 0704-0188

1. REPORT DATE	2. REPORT TYPE	3. DATES COVERED
19 MAR 1986	**Technical Memo**	**19-03-1986 to 19-03-1986**

4. TITLE AND SUBTITLE **Computation of Complex Airy Functions**	5a. CONTRACT NUMBER
	5b. GRANT NUMBER
	5c. PROGRAM ELEMENT NUMBER
6. AUTHOR(S) **Marvin Goldstein**	5d. PROJECT NUMBER **W65000**
	5e. TASK NUMBER
	5f. WORK UNIT NUMBER
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) **Naval Underwater Systems Center,New London,CT,06320**	8. PERFORMING ORGANIZATION REPORT NUMBER **TM No. 861032**
9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)	10. SPONSOR/MONITOR'S ACRONYM(S)
	11. SPONSOR/MONITOR'S REPORT NUMBER(S)

12. DISTRIBUTION/AVAILABILITY STATEMENT
Approved for public release; distribution unlimited

13. SUPPLEMENTARY NOTES
NUWC2015

14. ABSTRACT
In this paper, we discuss an efficient way to evaluate scaled complex Airy functions by asymptotic series. By obtaining an a-priori estimate of the number of monotonically decreasing terms of the asymptotic series for exp (z^1.5/1.5)Ai(z) that can be summed without underflowing the unit round-off of the computer arithmetic, we avoid unnecessary computations during the summation. Using this method, we develop an efficient computational routine for obtaining numerically linearly independent Airy functions whose Wronskian is stable throughout the complex plane. The scaled complex Airy function of the second kind exp (-z^1.5)Bi(z) may be obtained with well-known connecting formulas from the numerically independent computations. However, the Wronskian of the Airy functions of the first and second kind is not stable in the sector 180??> abs(arg z) > 60??, when abs(z) is sufficiently large.

15. SUBJECT TERMS
Airy Function; Wrongskian

16. SECURITY CLASSIFICATION OF:			17. LIMITATION OF ABSTRACT	18. NUMBER OF PAGES	19a. NAME OF RESPONSIBLE PERSON
a. REPORT **unclassified**	b. ABSTRACT **unclassified**	c. THIS PAGE **unclassified**	**Same as Report (SAR)**	**25**	

86-1032
CY001

REFERENCE ONLY

TM No. 861032

NAVAL UNDERWATER SYSTEMS CENTER
NEW LONDON LABORATORY
NEW LONDON, CONNECTICUT 06320

Technical Memorandum

COMPUTATION OF COMPLEX AIRY FUNCTIONS

Date: 19 March 1986

Prepared by: Marvin J. Goldstein
Marvin J. Goldstein
Surface Ship Sonar Dept

ABSTRACT

In this paper, we discuss an efficient way to evaluate scaled complex Airy functions by asymptotic series. By obtaining an a-priori estimate of the number of monotonically decreasing terms of the asymptotic series for $exp(z^{1.5}/1.5)Ai(z)$ that can be summed without underflowing the unit round-off of the computer arithmetic, we avoid unnecessary computations during the summation. Using this method, we develop an efficient computational routine for obtaining numerically linearly independent Airy functions whose Wronskian is stable throughout the complex plane. The scaled complex Airy function of the second kind $exp(-z^{1.5}/1.5)Bi(z)$ may be obtained with well-known connecting formulas from the numerically independent computations. However, the Wronskian of the Airy functions of the first and second kind is not stable in the sector $180° > abs(arg\ z) > 60°$, when $abs(z)$ is sufficiently large.

ADMINISTRATIVE INFORMATION

This memorandum was prepared under Job Order No. W65000, EVA Prog. Modeling Efforts, Principal Investigator, H. Weinberg (Code 3332). The author of this memorandum is located at the New London Laboratory, Naval Underwater Systems Center, New London, Ct. 06320.

INTRODUCTION

Typically the asymptotic expansion of the scaled Airy function *exp(zta)Ai(z)* $(zta=z^{1.5}/1.5)$ is truncated to obtain a function that approximates *exp(zta)Ai(z)* with arbitrary accuracy for sufficiently large values of abs(z). On the other hand, for any specific value of z there is a limit to the accuracy with which we can compute *exp(zta)Ai(z)* by truncating its asymptotic expansion, and the accuracy cannot always be improved by taking additional terms of the asymptotic expansion. Accuracy may be improved by adding additional terms as long as the magnitude of the terms continue to decrease without underflowing the computer's unit round-off error. Although Thomson [1] has implemented a complex Airy function computer code, the code runs slowly because comparison tests are performed to assure monotonicity and avoid the accumulation of terms that underflow the smallest positive floating-point computer number instead of the computer's larger unit round-off error.

In this paper, we discuss an efficient way to evaluate scaled complex Airy functions by asymptotic series. By obtaining an a-priori estimate of the number of monotonically decreasing terms of the asymptotic series for *exp(zta)Ai(z)* that can be summed without underflowing the unit round-off of the computer arithmetic, we avoid unnecessary computations during the summation. Using this method, we develop an efficient computational routine for obtaining numerically linearly independent Airy functions whose Wronskian is stable throughout the complex plane. The scaled complex Airy function of the second kind *exp(-zta)Bi(z)* may be obtained with well-known connecting formulas from the numerically independent computations. However, the Wronskian of the Airy functions of the first and second kind is not stable in the sector $180^{\circ} > abs(\arg z) > 60^{\circ}$, when *abs(z)* is sufficiently large.

ASYMPTOTIC ANALYSIS

Consider approximating the scaled Airy function $exp(zta)Ai(z)$ for values of z of large magnitude by evaluating a partial sum of the asymptotic series [2]

$$exp(zta)Ai(z) \sim (\pi^{-0.5}z^{-0.25}/2)\sum_{k=0}^{\infty}(-1)^k c_k zta^{-k} \quad (1)$$

where

$$zta = z^{1.5}/1.5, \quad abs(\ arg\ z\ ) < \pi$$

$$c_0 = 1, \quad c_{k+1}/c_k = k/2 + 5(k+1)^{-1}/72. \quad (2)$$

Writing

$$G_k = (-1)^k c_k zta^{-k}$$

we see that the terms of the sum in (1) can be generated recursively:

$$G_{k+1} = -(c_{k+1}/c_k)zta^{-1}G_k \quad (k = 0,1,2, \ ...) \quad (3)$$

and, a fortiori,

$$abs(G_{k+1}) = abs(zta^{-k-1})\prod_{j=0}^{k}(c_{j+1}/c_j)$$
$$\doteq 2^{-k}k!\,abs(zta^{-k-1})5/72 \equiv abs(\hat{G}_{k+1}). \quad (4)$$

Let

$$S_{n-1}(z) = \sum_{k=0}^{n-1}G_k \ . \quad (5)$$

By definition of an asymptotic series [3], as $z \to \infty$

$$\lim zta^n(2\pi^{.5}z^{.25}exp(zta)Ai(z)-S_{n-1}(z))=(-1)^n c_n \ .$$

Therefore, for values of z of sufficiently large magnitude

$$2\pi^{0.5}z^{0.25}exp(zta)Ai(z) - S_{n-1}(z) \doteq G_n. \ (6)$$

For a given value of z, if G_n is the first term in series $S_\infty(z)$ for which

$$abs(G_{n+1}) > abs(G_n) \quad (7)$$

then the magnitude of the error in $S_{n-1}(z)$ is "minimized", since G_n will have the smallest magnitude of all the terms in the series. By (3), condition (7) is equivalent to

$$c_{n+1}/c_n > m = abs(zta). \qquad (8)$$

Therefore, the first value of n for which (7) holds is

$$n = [(9m^2 + 9m + 1)^{0.5}/3 + m - 0.5] \sim [2m - (14.4m)^{-1}] \qquad (9)$$

where [x] is the smallest integer greater than or equal to x. However, if m is sufficiently large, the computation of G_k will cause computer underflow well before k assumes the value of n given in (9), and the evaluation of (5) out to G_{n-1} then becomes impractical. In this case, (5) could be summed until $abs(G_k)$ underflows the smallest positive computer number s. In [4], it was shown how to accomplish this without executing comparison tests on G_k, resulting in a 40% improvement in execution efficiency. However, an additional 50% improvement in execution efficiency is possible by considering the asymptotic behavior of the relative error in the approximation $S_{n-1}(z)$ to

$$2\pi^{0.5}z^{0.25}exp(zta)Ai(z). \qquad (10)$$

Since

$$S_{n-1}(z) - 2\pi^{0.5}z^{0.25}exp(zta)Ai(z) \rightarrow 0 \text{ as } z \rightarrow \infty \quad (11)$$

it follows that

$$2\pi^{0.5}z^{0.25}exp(zta)Ai(z) \rightarrow 1 \text{ as } z \rightarrow \infty \qquad (12)$$

and, a fortiori, for the relative error in $S_{n-1}(z)$ we have

$$\frac{S_{n-1}(z) - 2\pi^{0.5}z^{0.25}exp(zta)Ai(z)}{2\pi^{0.5}z^{0.25}exp(zta)Ai(z)} \longrightarrow (S_{n-1}(z) - 1)/1$$

$$\longrightarrow 0/1 \qquad (13)$$

as $z \rightarrow \infty$. But the difference between a function and any partial sum of its asymptotic series is of the order of the first neglected term of the series as $z \rightarrow \infty$. Therefore, (13) suggests that for sufficiently large

values of $abs(z)$, we should be able to approximate (10) almost to the relative precision of the computer arithmetic by summing (5) until $abs(G_k)$ is smaller than the computer's unit round-off error μ instead of the computer's smallest positive number s. For a practical computer evaluation of $S_{n-1}(z)$ that avoids comparison tests on G_k, we develop an a-priori estimate that depends on z and μ of the number of monotonically decreasing terms of $S_{n-1}(z)$ that can be summed without underflowing the computer's unit round-off error μ. To do this, we proceed in essentially the same way we did in [4], replacing the smallest positive computer number s with the computer unit round-off error μ.

An <u>estimate</u> of the largest value of $m=abs(zta)$ for which the magnitude of the terms of the partial sum $S_n(z)$ is monotonically decreasing and

$$\mu < abs(G_n) < abs(G_{n+1}) = \mu \prod_{j=1}^{n} (1 + (j+1)^{-1}/(7.2j)) \qquad (14)$$

can be obtained by setting approximation (4) for $abs(G_{n+1})$ equal to μ with $n=2m$, applying Stirling's asymptotic formula for factorials and solving the resulting equation for m. This gives m^* as the largest value of m, where m^* is the limit of the rapidly convergent sequence

$$m_j=-ln(m_{j-1})/4 + m_0, \quad m_0=-ln(7.2\mu\pi^{-0.5})/2 \qquad (15)$$

In other words, for any value of z for which $m<m^*$, the theoretical "minimal" error in $S_{n-1}(z)$ (*namely* $abs(G_n)$ *with* $n=2m$) is computationally attainable, because it exceeds μ. On the other hand, for any value of z for which $m>m^*$, the theoretical "minimal" error in $S_{n-1}(z)$ will underflow μ or be close to μ,

since it satisfies

$$abs(G_n(m)) < abs(G_k(m)) \qquad (k = 2m^*)$$

$$\doteq abs(\hat{G}_k(m))$$

$$= abs(\hat{G}_{k+1}(m^*))(m/m^*)^{-k}, \text{ by } (4)$$

$$= \mu(m/m^*)^{-k} \to 0 \text{ as } m/m^* \to \infty. \quad (16)$$

Therefore, for values of z for which $m>m^*$, we can approximate (10) almost to the relative precision of the computer arithmetic by summing no more than $k=2m^*$ terms of $S_{n-1}(z)$; in fact, we evaluate $S_{[k]-p-1}(z)$ for a positive integer value of $p<[k]$, in order to avoid computing terms of $S_{[k]-1}(z)$ that underflow μ when m/m^* is large.[1]

To select $S_{[k]-p-1}(z)$ when $m>m^*$, we determine the index $[k]-p$ $(=[k-p])$ of the error term

$$abs(G_{[k]-p}(m)) = abs(\hat{G}_{[k]-p}(m)) \prod_{j=1}^{[k]-p-1} (1 + (j+1)^{-1}/(7.2j))$$

for which

$$\mu < abs(\hat{G}_{[k]-(p+1)}(m)) > abs(\hat{G}_{[k]-p}(m)) \leq \mu \quad (17)$$

$$(p < [k], \; k = 2m^*)$$

in the following manner.

---

[1] If $\mu=2^{-60}$, then eq.(15) gives m^*=19.35 (abs(z)=9.44); therefore, for abs(z) > 9.44, no more than 39 terms of $S_{n-1}(z)$ are required to approximate (10) to the relative precision of the computer arithmetic. The number of required terms falls off at a moderate rate as m/m^* moves away from 1; e.g. for m/m^*=2, 5.7 (abs(z)=15, 30), the number of terms required is 15, 10, respectively. A list of the required number of terms $[2m^*-p]$ for corresponding values of m and m/m^* appears in Table 1.

Obtain a table of values of m, *namely* m_p, for positive integer values of $p<[k]$, which satisfy the equation

$$abs(\hat{G}_{[k]-p}(m)) = \mu . \qquad (18)$$

For a given value of $m>m^*$, *say* $m = \tilde{m}$, *if* $m_p<\tilde{m}\leq m_{p+1}$,

then

$$\mu = abs(\hat{G}_{[k]-p}(m_p)) > abs(\hat{G}_{[k]-p}(\tilde{m})) \qquad (19)$$

and

$$\mu = abs(\hat{G}_{[k]-(p+1)}(m_{p+1})) \leq abs(\hat{G}_{[k]-p-1}(\tilde{m})) \qquad (20)$$

since

$$abs(\hat{G}_{[k]-p}(m)) \begin{cases} \geq \mu, & \text{if } m \leq m_p \\ < \mu, & \text{if } m > m_p . \end{cases}$$

Therefore, if $m_p<\tilde{m}\leq m_{p+1}$ we evaluate $S_{[k]-(p+1)}(z)$, since $\hat{G}_{[k]-p}(\tilde{m})$ satisfies (17), by (19) and (20). For example, assuming $\mu=2^{-60}$, if a given value of $m>m^*$, *say* $m=19.6$, lies between the values of m for p=5,6 in Table 1, then take $[2m^*-5]=34$ *for* $[k-p]$.

p	m	m/m*	[2m* - p]
0	19.35258	1	39
1	19.36194	1.000484	38
2	19.38579	1.001716	37
3	19.42577	1.003782	36
4	19.48381	1.006781	35
5	19.5621	1.010827	34
6	19.66328	1.016055	33
7	19.79037	1.022622	32
8	19.94698	1.030714	31
9	20.1374	1.040554	30
10	20.36676	1.052406	29
11	20.64125	1.066589	28
12	20.96842	1.083495	27
13	21.35754	1.103602	26
14	21.8201	1.127504	25
15	22.37055	1.155947	24
16	23.02719	1.189877	23
17	23.81359	1.230513	22
18	24.76054	1.279444	21
19	25.90883	1.338779	20
20	27.31371	1.411373	19
21	29.05135	1.501162	18
22	31.22937	1.613706	17
23	34.00401	1.757079	16
24	37.60926	1.943372	15
25	42.4084	2.191357	14
26	48.99014	2.531453	13
27	58.35861	3.015547	12
28	72.336	3.737796	11
29	94.49151	4.882632	10
30	132.5262	6.847986	9
31	205.2731	10.60702	8
32	367.2325	18.97589	7
33	818.2973	42.28363	6
34	2605.586	134.6377	5
35	15653.95	808.882	4
36	342085.9	17676.5	3
37	2.000795E+08	1.033865E+07	2
38	8.006342E+16	4.137093E+15	1

Number of terms needed to obtain exp(zta) Ai(z) to a relative precision of $u = 2^{-60}$ for values of $m > m^* = 19.35$ is $[2m^* - p]$

TABLE 1

Analyzing the asymptotic series for $exp(zta)Ai'(z)$:

$$exp(zta)Ai'(z) \sim -(\pi^{-0.5}z^{0.25}/2)\sum_{k=0}^{\infty}(-1)^k d_k zta^{-k} \quad (21)$$

where

$$zta = z^{1.5}/1.5, \quad abs(\ arg\ z\ ) < \pi$$

$$d_0 = 1, \quad d_{k+1}/d_k = k/2 - 7(k+1)^{-1}/72 \qquad (22)$$

in the same way we analyzed (1) produces similar results. Writing

$$G_k = (-1)^k d_k zta^{-k}$$

we see that the terms of the sum in (21) can be generated recursively:

$$G_{k+1} = -(d_{k+1}/d_k)zta^{-1}G_k \quad (k = 0,1,2,\ ...) \quad (23)$$

and, a fortiori,

$$abs(G_{k+1}) = abs(zta^{-k-1})(7/72)\prod_{j=1}^{k}(d_{j+1}/d_j)$$
$$\doteq 2^{-k}k!\,abs(zta^{-k-1})7/72 \equiv abs(\hat{G}_{k+1}) \qquad (24)$$

Let

$$S_{n-1}(z) = \sum_{k=0}^{n-1}G_k \qquad (25)$$

For a given value of z, the magnitude of the theoretical error in $S_{n-1}(z)$ (namely $abs(G_n)$) is minimized when

$$n = [2m + 7/(12m)] \qquad (26)$$

An estimate of the largest value of m for which the minimal error satisfies:

$$abs(G_n) = \mu'\prod_{j=1}^{n-1}(\ 1 - 14(j + 1)^{-1}/(72j)\ ) \qquad (\mu' = 7\mu/5)$$

can be obtained by setting approximation(24) for $abs(G_n)$ equal to μ' with $n-1=2m$, applying Stirling's asymptotic formula for factorials and solving the resulting equation for m. This gives m^* as the largest value of m, where m^* is the limit of the rapidly convergent sequence

$$m_j=-ln(m_{j-1})/4 + m_0, \quad m_0=-ln(7.2\mu\pi^{-0.5})/2$$

Therefore, for values of z for which $m > m^*$, we can approximate

$$-(2\pi^{0.5}z^{-0.25})\exp(z\,ta)Ai'(z). \qquad (27)$$

to the relative precision of the computer arithmetic by summing no more than $k=2m^*+1$ terms of $S_{n-1}(z)$; in fact, for a positive integer value of $p<[k]$, we evaluate $S_{[k]-p-1}(z)$, where $[k]-p$ is the index of the error term

$$abs(G_{[k]-p}(m)) = abs(\hat{G}_{[k]-p}(m)) \prod_{j=1}^{[k]-p-1} (1 - 14(j+1)^{-1}/(72j))$$

that satisfies

$$\mu' < abs(\hat{G}_{[k]-(p+1)}(m)) > abs(\hat{G}_{[k]-p}(m)) \le \mu' \quad (28)$$

$$(p < [k],\ k = 2m^* + 1).$$

The index $[k]-p$ is determined in the following manner.

Since $\mu'=7\mu/5$, the table of values of m for which approximation(24) reduces to μ' is identical to Table 1 for $\mu=2^{-60}$. Therefore, for a given value of $m>m^*$, we determine $[k]-p$ $(=[2m^*]-p+1)$ by locating where m falls in Table 1. For example, if a given value of $m>m^*$, say $m=19.6$, lies between p=5,6 in Table 1, take $[2m^*-4]=35$ for $[k]-p$ $(=[2m^*]-p+1)$.

COMPUTATIONAL STABILITY OF THE WRONSKIAN

As pointed out in [5], in the sector $60° < abs(\arg z) < 180°$

$$Ai(z) \sim \mp iBi(z), \quad Ai'(z) \sim \mp iBi(z) \quad (abs(z)\ large) \quad (29)$$

Therefore, although theoretically the Wronskian $W(Ai(z),\ Bi(z))$ satisfies

$$W(Ai(z),\ Bi(z)) = \pi^{-1}, \quad (30)$$

it is numerically unstable in this sector when $abs(z)$ is large because of severe cancellation of significant digits; in other words, computationally $W(Ai(z),Bi(z))$ may be close to zero. For example, if $abs(z)=15$ and $arg\ z=150°$, then (to at least 20 significant digits)

	Real Part	*Imaginary Part*
$Ai(z)$:	-3010621074.950100596341	112308057590.0163025659
$Bi(z)$:	-112308057590.0163025659	-3010621074.950100596341
$Ai'(z)$:	422189739015.1793969374	-99699231497.52144586569
$Bi'(z)$:	99699231497.52144586569	422189739015.1793969374

so that

$$Ai(z) = -iBi(z), \quad Ai'(z) = -iBi'(z) \quad (31)$$

and $W(Ai(z),\ Bi(z))=0$ on computer aithmetic hardware having 20 significant decimal digits or less. On the other hand, this instability does not occur for the following numerically linearly independent Airy functions

$$Ai(z),\ Ai(z\ exp(\pm 2\pi i/3)) \quad (32)$$

where

$$W(Ai(z),\ Ai(z\ exp(2\pi i/3)) = 0.5\pi^{-1}exp(-\pi i/6) \quad (Im(z) < 0)$$

$$W(Ai(z),\ Ai(z\ exp(-2\pi i/3)) = 0.5\pi^{-1}exp(\pi i/6) \quad (Im(z) \geq 0)$$

COMPUTER PROGRAM

For $m > 19.35$ ($abs(z) > 9.44$), the complex Airy function subprogram computes the scaled numerically linearly independent Airy functions:

$$exp(zta)\ Ai(z),\quad exp(-zta)\ Ai(z\ exp(\mp 2\pi i/3)) \qquad (33)$$

and their scaled derivatives

$$exp(zta)\ Ai'(z),\quad exp(-zta)\ Ai'(z\ exp(\mp 2\pi i/3))exp(\mp 2\pi i/3) \qquad (34)$$

by evaluating partial sums of their asymptotic expansions (1, 21) to the working precision of a computer with unit round-off error $\mu=2^{-60}$, avoiding the accumulation of terms that underflow μ. In evaluating

$$exp(-zta)\ Ai(z\ exp(\mp 2\pi i/3)),\ exp(-zta)\ Ai'(z\ exp(\mp 2\pi i/3))exp(\mp 2\pi i/3)$$

we rotate z $120°$ counter-clockwise if its imaginary part is negative; otherwise z is rotated $120°$ *clockwise away from* $180°$.

For $7.45 < m \leq 19.35$ ($5 < abs(z) \leq 9.44$), we evaluate only $[2m+1]$ terms of the asymptotic expansions (1, 21), so that the largest error occurs for $m=7.45$

For $0 \leq abs(z) \leq 5$, we employ the ascending power series given in [2] to compute the Airy functions of the first and second kind from which we obtain

$$exp(0)\ Ai(z\ exp(\pm 2\pi i/3)) = 0.5exp(\mp \pi i/6)\ (Bi(z) \pm iAi(z))$$

$$exp(0)\ Ai'(z\ exp(\pm 2\pi i/3))exp(\pm 2\pi i/3) = 0.5exp(\mp \pi i/6)\ (Bi'(z) \pm iAi'(z))$$

The scaled complex Airy function of the second kind:

$$exp(-zta)Bi(z), \quad exp(-zta)Bi'(z) \qquad (abs(\ arg\ z\ ) \leq 60°) \qquad (35)$$

$$exp(zta)Bi(z), \quad exp(zta)Bi'(z) \qquad (60° < abs(\ arg\ z\ ) \leq 180°)$$

may be obtained in the asymptotic region $(abs(z)>5)$ from the numerically linearly independent scaled computations by utilizing the connection formulas:

$$Bi(z) = 2exp(\pm\pi i/6)\ Ai(z\ exp(\pm 2\pi i/3)) \mp iAi(z) \qquad (36)$$

$$Bi'(z) = 2exp(\pm\pi i/6)\ Ai'(z\ exp(\pm 2\pi i/3))exp(\pm 2\pi i/3) \mp iAi'(z)$$

If $abs(\ arg\ z\ )\leq 60°$, then $Real(zta)\geq 0$, so we compute

$$exp(-zta)Bi(z) = 2exp(\pm\pi i/6)\ (\ exp(-zta)Ai(z\ exp(\pm 2\pi i/3))\ )$$
$$\mp iexp(-2zta)\ (\ exp(zta)Ai(z)\ ) \qquad (37)$$

$$exp(-zta)Bi'(z) = 2exp(\pm 5\pi i/6)\ (\ exp(-zta)\ Ai'(z\ exp(\pm 2\pi i/3))\ )$$
$$\mp\ iexp(-2zta)\ (\ exp(zta)Ai'(z)\ ),$$

where the upper sign is selected if $Im(z)<0$, and the lower sign, if $Im(z)\geq 0$. If $60°<abs(\ arg\ z\ )\leq 180°$, then $Real(zta)<0$, so we compute

$$exp(zta)Bi(z) = 2exp(\pm\pi i/6)\ exp(2zta)\ (\ exp(-zta)Ai(z\ exp(\pm 2\pi i/3))\ )$$
$$\mp\ i(\ exp(zta)Ai(z)\ ) \qquad (38)$$

$$exp(zta)Bi'(z) = 2exp(\pm 5\pi i/6)\ exp(2zta)\ (\ exp(-zta)Ai'(z\ exp(\pm 2\pi i/3))\ )$$
$$\mp\ i(\ exp(zta)Ai'(z)\ ),$$

where again the upper sign is selected if $Im(z)<0$, and the lower sign, if $Im(z)\geq 0$.

REFERENCES

1. D. J. Thomson, "Implementation of the Lamont-Doherty Geological Observatory Normal Mode/Fast Field Model on the NUSC VAX 11/780 Computer," NUSC TR 7221, 18 September 1984.

2. M. Abramowitz and I. Stegun, Handbook of Mathematical Functions with Formulas, Graphs and Mathematical Tables, NBS Applied Math. Series 55, 1964.

3. M. R. Spiegel, Theory and Problems of Complex Variables, Schaum Publishing Co., New York, 1964.

4. M. J. Goldstein, "Mathematically Equivalent, Computationally Non-equivalent, Formulas and Software Comprehensibility," NUSC TM. 851062, May 1985.

5. Z. Schulten, D. Anderson, R. Gordon, "An Algorithm for the Evaluation of the Complex Airy Functions," in Journal of Computational Physics 31, 60-75 (1979)

```
C *****************************************************************
C *  IMPROVED COMPLEX DOUBLE PRECISION AIRY FUNCTION...M. GOLDSTEIN *
C *                                                                 *
C *                                                                 *
C *  GIVEN Z, IF INDEP > 0, CLAIRY RETURNS THE NUMERICALLY LINEARLY *
C *  INDEPENDENT AIRY FUNCTIONS:                                    *
C *                                                                 *
C *           BI = EXP(-ZTAM) * Ai( Z*EXP(-/+i2PI/3) )              *
C *                              '                                  *
C *      BIP = EXP(-ZTAM) * Ai( Z*EXP(-/+i2PI/3) ) * EXP(-/+i2PI/3) *
C *                                                '                *
C *           ( AI, AIP ) = EXP(ZTAM) * ( Ai(Z), Ai(Z) )            *
C *                                                                 *
C *                 ( ZTAM = 2 * SQRT(Z)**3 / 3 )                   *
C *                                                                 *
C *  FOR THE COMPUTATION OF BI AND BIP, Z IS ROTATED 120 DEGREES    *
C *  COUNTER-CLOCKWISE IF ITS IMAGINARY PART IS NEGATIVE; OTHER-    *
C *  WISE Z IS ROTATED 120 DEGREES CLOCKWISE.  THE WRONSKIAN OF THE *
C *  NUMERICALLY INDEPENDENT AIRY FUNCTIONS IS STABLE THROUGHOUT THE*
C *  COMPLEX PLANE.                                                 *
C *                                                                 *
C *                                                                 *
C *  GIVEN Z, IF INDEP < 0, CLAIRY RETURNS THE FIRST AND 2ND KIND   *
C *  AIRY FUNCTIONS:                                                *
C *                                             '                   *
C *        ( AI, AIP ) = EXP(ZTAM) * ( Ai(Z), Ai(Z) )               *
C *                                      '                          *
C *  ( BI, BIP ) = EXP(-ZTAM) * ( Bi(Z), Bi(Z) ), IF REAL(ZTAM) >=0 *
C *                                      '                          *
C *              = EXP(ZTAM) * ( Bi(Z), Bi(Z) ), IF REAL(ZTAM) < 0  *
C *                          '                 '                    *
C *  SINCE Ai(Z) ~ -/+iBi(Z) AND Ai(Z) ~ -/+iBi(Z) IF 60 < ABS(ARG Z)
C *  <  180, THE WRONSKIAN OF THE AIRY FUNCTIONS OF THE FIRST AND   *
C *  SECOND KIND IS NOT STABLE THROUGHOUT THE COMPLEX PLANE.        *
C *                                                                 *
C *****************************************************************
      SUBROUTINE CLAIRY(Z, AI, BI, AIP, BIP, ZTAM, INDEP )
      IMPLICIT REAL*8 ( A-H, O-Z )
      COMPLEX*16  ZS, ZSS, ZTA, SC1, SC2, SD1, SD2, SQM1, TSQM1,
     +   HEPI06, CHEP06, EPI06, CEPI06, E2PI03, CE2P03, TEPI06, TCPI06,
     +   T1, T2, CK, ZSQ, F, G, Q, R, ZB, ZERO, ZO, REPI06, E2ZTAM,
     +   Z, AI, BI, AIP, BIP, ZTAM
      DIMENSION   C(40), D(40)
      DATA RDEG/.1745329251994329D-1/, PI04/.7853981633974483D0/
      DATA RC2RPI/.2820947917738781D0/, RPI/1.772453850905516D0/
      DATA SQR3/1.732050807568877290/,ZERO/(0.D0,0.D0)/
      DATA SQM1/(0.0D0, 1.0D0)/, TSQM1/(0.0D0, 2.0D0)/
      DATA SN60,CS60/.8660254037844386470D0, 0.5D0/,
     +     EPI06/(.8660254037844386470D0, 0.5D0)/,
     +     CEPI06/(.8660254037844386470D0, -0.5D0)/,
     +     TEPI06/(1.7320508075688772900D0, 1.0D0)/,
     +     TCPI06/(1.7320508075688772900D0, -1.0D0)/,
     +     E2PI03/(-0.5D0, .8660254037844386470D0)/,
     +     CE2P03/(-0.5D0, -.8660254037844386470D0)/,
     +     HEPI06/(.4330127018922193233D0, 0.25D0)/,
     +     CHEP06/(.4330127018922193233D0, -0.25D0)/
      DATA C/    1.D0,              0.6944444444444444D-01,
     +   0.5347222222222222D+00,   0.1023148148148148D+01,
     +   0.1517361111111111D+01,   0.2013888888888888D+01,
     +   0.2511574074074074D+01,   0.3009920634920634D+01,
```

```
     +    0.35086805555555555D+01,   0.40077160493827160D+01,
     +    0.45069444444444444D+01,   0.50063131313131313D+01,
     +    0.55057870370370370D+01,   0.60053418803418803D+01,
     +    0.65049603174603174D+01,   0.70046296296296296D+01,
     +    0.75043402777777777D+01,   0.80040849673202614D+01,
     +    0.85038580246913580D+01,   0.90036549707602339D+01,
     +    0.95034722222222223D+01,   0.10003306878306878D+02,
     +    0.10503156565656565D+02,   0.11003019323671497D+02,
     +    0.11502893518518518D+02,   0.12002777777777777D+02,
     +    0.12502670940170940D+02,   0.13002572016460905D+02,
     +    0.13502480158730158D+02,   0.14002394636015325D+02,
     +    0.14502314814814814D+02,   0.15002240143369175D+02,
     +    0.15502170138888888D+02,   0.16002104377104377D+02,
     +    0.16502042483660130D+02,   0.17001984126984126D+02,
     +    0.17501929012345679D+02,   0.18001876876876877D+02,
     +    0.18501827485380117D+02,   0.19001780626780626D+02/
C
      DATA D/       1.D0,           -0.97222222222222222D-01,
     +    0.45138888888888888D+00,   0.96759259259259259D+00,
     +    0.14756944444444444D+01,   0.19805555555555555D+01,
     +    0.24837962962962962D+01,   0.29861111111111111D+01,
     +    0.34878472222222222D+01,   0.39891975308641975D+01,
     +    0.44902777777777777D+01,   0.49911616161616161D+01,
     +    0.54918981481481482D+01,   0.59925213675213675D+01,
     +    0.64930555555555555D+01,   0.69935185185185185D+01,
     +    0.74939236111111111D+01,   0.79942810457516340D+01,
     +    0.84945987654320987D+01,   0.89948830409356725D+01,
     +    0.94951388888888888D+01,   0.99953703703703704D+01,
     +    0.10495580808080808D+02,   0.10995772946859903D+02,
     +    0.11495949074074074D+02,   0.11996111111111111D+02,
     +    0.12496260683760683D+02,   0.12996399176954732D+02,
     +    0.13496527777777777D+02,   0.13996647509578544D+02,
     +    0.14496759259259259D+02,   0.14996863799283154D+02,
     +    0.15496961805555555D+02,   0.15997053872053872D+02,
     +    0.16497140522875817D+02,   0.16997222222222222D+02,
     +    0.17497299382716049D+02,   0.17997372372372372D+02,
     +    0.18497441520467836D+02,   0.18997507122507122D+02/
C
      ZTAM = ZERO
      IF (CDABS(Z) .GT. 5.) GO TO 200
C     POWER SERIES EXPANSION OF AIRY FUNCTION FOR ABS(Z).LE.5
      T1   =  DCMPLX(.35502805388781723900,0.D0)
      T2   =  DCMPLX(.25881940379280679D0,0.D0)
      F    =  T1
      R    =  T2
      T2   =  T2*Z
      G    =  T2
      ZSQ  =  Z*Z
      Q    =  ZERO
      XN   =  1.D0
C
  100 XN   =  XN+1.D0
      T1   =  T1*ZSQ/XN
      Q    =  Q+T1
C
      XN   =  XN+1.D0
      T1   =  T1*Z/XN
      CK   =  F
      F    =  F+T1
C
```

```
      T2    =  T2*ZSQ/XN
      R     =  R+T2
C
      XN    =  XN+1.D0
      T2    =  T2*Z/XN
      G     =  G+T2
C
      IF ( DREAL(CK).NE. DREAL(F).OR.DIMAG(CK).NE.DIMAG(F))  GO TO 100
C
      AI  = F-G
      AIP = Q-R
      BI  = SQR3*(F+G)
      BIP = SQR3*(Q+R)
      IF( INDEP .GT. 0 ) THEN   ! SCALED INDEPENDENT BIRY, AIRY
         IF( DIMAG(Z) .GE. 0.0D0 ) THEN
            BI = HEPIO6 * ( BI - SQM1 * AI )
            BIP = HEPIO6 * ( BIP - SQM1 * AIP )
         ELSE
            BI = CHEPO6 * ( BI + SQM1 * AI )
            BIP = CHEPO6 * ( BIP + SQM1 * AIP )
         END IF
      END IF
      RETURN
C
C     ASYMPTOTIC EXPANSION OF AIRY FUNCTION FOR LARGE Z.
C     GET ABS( ARG Z )
  200 TH = DABS( DATAN2( DIMAG(Z), DREAL(Z) ) / RDEG )
      T2 = E2PIO3
      IF(DIMAG(Z) .GE. 0.0D0) T2 = CE2PO3
C
      IF(TH .GT. 150.0D0) THEN
         ZO = Z * T2                ! ROTATE Z
         ZS = CDSQRT(ZO)
         ZSS = CDSQRT(ZS)
         ZTA = ZO * ZS / 1.5D0
         ZTAM = -ZTA
         E2ZTAM = ZERO
         IF ( DREAL(ZTA) .LE. 43.D0 ) E2ZTAM = CDEXP(2.D0 * ZTAM)
         CALL SUMMER(ZTA, SC1, SC2, C, 0)
         CALL SUMMER(ZTA, SD1, SD2, D, 1)
         BI = RC2RPI / ZSS * SC1         ! EXP(ZTA)Ai(ZO)
         BIP = -RC2RPI * ZSS * SD1 * T2  ! EXP(ZTA)Aip(ZO)T2
         IF(DIMAG(Z) .GE. 0.0D0) THEN
            AI = CHEPO6 * (SC2/(RPI*ZSS) + (TSQM1*E2ZTAM)*BI)
            AIP = CHEPO6 * (SD2/RPI*ZSS*T2 + (TSQM1*E2ZTAM)*BIP)
         ELSE
            AI = HEPIO6 * (SC2/(RPI*ZSS) - (TSQM1*E2ZTAM)*BI)
            AIP = HEPIO6 * (SD2/RPI*ZSS*T2 - (TSQM1*E2ZTAM)*BIP)
         END IF
      ELSE                            ! FOR ABS( ARG Z) .LE. 150 DEGREES
C                                   USE NBS 10.4.59, 10.4.61 FOR AI, AIP
         ZS = CDSQRT(Z)
         ZSS = CDSQRT(ZS)
         ZTA = Z*ZS/1.5D0
         CALL SUMMER (ZTA, SC1, SC2, C, 0)
         ZTAM = ZTA
         AI = RC2RPI/ZSS*SC1
         CALL SUMMER (ZTA, SD1, SD2, D, 1)
         AIP = -RC2RPI*ZSS*SD1
         REPIO6 = EPIO6
```

```
         IF( DIMAG(Z) .GE. 0.0D0 ) REPIO6 = CEPIO6
         BIP = -RC2RPI * SD2 * ZSS * REPIO6 * T2
         BI = RC2RPI * SC2 / (ZSS * REPIO6)
      END IF
C
      IF( INDEP .LT. 0) THEN    ! SCALED AIRY OF 2ND KIND
         E2ZTAM = 0.0D0
         IF( DREAL(ZTAM) .GE. 0.0D0 ) THEN     ! ABS( ARG Z ) <= 60
            IF( DREAL(ZTAM) .LE. 43.D0 ) E2ZTAM = CDEXP(-2.D0 * ZTAM)
            IF( DIMAG(Z) .GE. 0.0D0 ) THEN
               BI = SC2 / (ZSS*RPI) + (SQM1 * E2ZTAM) * AI
               BIP = SD2 * ZSS / RPI + (SQM1 * E2ZTAM) * AIP
            ELSE
               BI = SC2 / (ZSS*RPI) - (SQM1 * E2ZTAM) * AI
               BIP = SD2 * ZSS / RPI - (SQM1 * E2ZTAM) * AIP
            END IF
         ELSE                             ! 60 < ABS( ARG Z ) <= 180
            IF( DREAL(ZTAM) .GE. -43.D0) E2ZTAM = CDEXP(2.D0 * ZTAM)
            IF( DIMAG(Z) .GE. 0.0D0 ) THEN
               BI = (TCPIO6 * E2ZTAM) * BI + SQM1 * AI
               BIP = (TCPIO6 * E2ZTAM) *BIP + SQM1 * AIP
            ELSE
               BI = (TEPIO6 * E2ZTAM) * BI - SQM1 * AI
               BIP = (TEPIO6 * E2ZTAM) * BIP - SQM1 * AIP
            END IF
         END IF
      END IF
      RETURN
      END
```

```fortran
      SUBROUTINE SUMMER (ZTA, S1, S2, CF, NGATE)
      DOUBLE PRECISION CF
      REAL MSTAR
      COMPLEX*16 ZTA,VZTA,G,S1,S2
      DIMENSION CF(40)
      DATA MSTAR/19.35/
C                               MSTAR IS THE LIMIT OF THE SEQUENCE:
C                               M(J) = -LN( M(J-1) ) / 4 + M(O)
C                               ( M(O) = -LN( 7.2 U / SQR(PI) ) / 2 )
C                               WHERE U IS THE COMPUTER'S UNIT ROUND-
C                               OFF ERROR (U = 2**(-60) HERE).
      IF(NGATE.EQ.0)THEN
         ABSZTA = CDABS(ZTA)
         IF(ABSZTA .GT. MSTAR)THEN
C                                    FETCH NUMBER OF TERMS FROM TABLE
            NTERMS = NTERM( ABSZTA )
         END IF
      END IF
      IF(ABSZTA .LE. MSTAR)THEN
         NTERMS = 2.0*ABSZTA + 2.0
         IF(SNGL(CF(NTERMS)) .GT. ABSZTA) NTERMS = NTERMS - 1
      END IF
      G = DCMPLX(CF(1),0.0D0)
      S1 = G
      S2 = G
      VZTA = G/ZTA
      DO 100 I = 2, NTERMS, 2
         G = G*(VZTA*CF(I))
         S1 = S1 - G
         S2 = S2 + G
         G = G*(VZTA*CF(I+1))
         S1 = S1 + G
         S2 = S2 + G
  100 CONTINUE
      IF( MOD(NTERMS, 2) .NE. 0 ) RETURN
      S1 = S1 - G
      S2 = S2 - G
      RETURN
      END
```

```
      FUNCTION NTERM(ABSZTA)
C   *****************************************************************
C   *** USING THE KEY ABSZTA, FUNCTION NTERM DOES A BINARY SEARCH ***
C   *** OF AZTA TO DETERMINE THE NUMBER OF TERMS OF THE SERIES TO ***
C   *** SUM.  THE TABLE OF AZTA VALUES WAS CONSTRUCTED FOR U =    ***
C   *** 2**(-60).  IF THE VALUE OF U CHANGES THEN A NEW TABLE OF  ***
C   *** AZTA VALUES SHOULD BE CONSTRUCTED AS DESCRIBED IN NUSC    ***
C   *** TM 861032.                                                ***
C   *****************************************************************
      INTEGER NTERM
      DIMENSION AZTA(38), NTABLE(38)
      DATA (AZTA(K), NTABLE(K), K = 1, 19)/
     +     19.35    ,  39,
     +     19.36194,  38,
     +     19.38579,  37,
     +     19.42577,  36,
     +     19.48381,  35,
     +     19.5621 ,  34,
     +     19.66328,  33,
     +     19.79037,  32,
     +     19.94698,  31,
     +     20.1374 ,  30,
     +     20.36676,  29,
     +     20.64125,  28,
     +     20.96842,  27,
     +     21.35754,  26,
     +     21.8201 ,  25,
     +     22.37055,  24,
     +     23.02719,  23,
     +     23.81359,  22,
     +     24.76054,  21/
      DATA (AZTA(K), NTABLE(K), K = 20, 38)/
     +     25.90883,  20,
     +     27.31371,  19,
     +     29.05135,  18,
     +     31.22937,  17,
     +     34.00401,  16,
     +     37.60926,  15,
     +     42.4084,   14,
     +     48.99014,  13,
     +     58.35861,  12,
     +     72.336  ,  11,
     +     94.49151,  10,
     +    132.5262 ,   9,
     +    205.2731 ,   8,
     +    367.2325 ,   7,
     +    818.2973,    6,
     +   2605.586  ,   5,
     +  15653.95   ,   4,
     +  342085.9   ,   3,
     +  342085.9   ,   3/
      I = 1
      J = 37
C     DO WHILE ( I .LE. J )
   20    IF( .NOT. (I .LE. J) ) GO TO 70
         L = ( I + J ) / 2
```

```
      IF( ABSZTA .LT. AZTA(L) ) THEN
         J = L - 1
      ELSE
         I = L + 1
      END IF
      GO TO 20
70 CONTINUE
   NTERM = NTABLE( I )
   RETURN
   END
```

COMPUTATION OF COMPLEX AIRY FUNCTIONS
TM 861032
Marvin J. Goldstein
Surface Ship Sonar Department
19 MARCH 1986
UNCLASSIFIED

DISTRIBUTION LIST

Code	
00	
01	E.L. Messere
01A	R. Moore
01Y	Dr. J. Short
01	D. Quigley
10	Dr. W. VonWinkle
101	F. Weigle
	Dr. L. Goodman
	Dr. K. Lima
02	Dr. C. Kindilien
0211	R. Bernier
021311	NL Library (3 copies)
021312	NPT Library
03	D. Walters
0303	G. Hill
20	W. Clearwaters
32	Dr. J. Kingsbury
3211	J. O'Sullivan
3211	A. Lesick
3211	Dr. N. Owsley
3212	S. Dzerovych
3212	Dr. J. Ianniello
3212	J. Pearson
3212	J. Ferrie
3213	S. Kessler
3213	W. Axtell
3213	R. Cox
3213	D. Rawson
3213	J. Sanchis
3232	G. Connolly
3232	Dr. R. Streit
3232	B. Helme, Jr.
3233	Dr. H. Schloemer
3233	Dr. S. Ko
3233	Dr. W. Strawderman
3234	D. T. Porter
325	W. Coggins
325	J. Shores
3251	D. Daros
3251	C. Bowman
3251	P. Miner
3252	T. Anderson
3253	J. Ionata
3253	M. Kuznitz
3253	J. Munoz
3253	R. Leask
3202	D. Yarger
3233	R. Christman
33	L. Freeman
3301	T. Bateman
33A	B. Cole
33A3	Dr. W. Roderick
33B	D. Counsellor
33C	J. Hanrahan
3312	C. Becker
3312	P. Saikowski
3313	J. Gregor
3313	D. Aker
3313	P. Anchors
3313	N. Suliniski
3313	D. Fingerman
3313	R. Molino
3314	Dr. C. Carter
3314	Dr. R. Dwyer
3314	Dr. A. Nuttall
3314	W. Goldman
3314	I. Cohen
3314	R. Johnson
3314	J. Sikorski
3314	Dr E. Eby
3314	R. Tremblay
332	Dr. R. Radlinski
3321	F. P. Fesseden
333	W. Schumacher
3331	D. Browning
3331	P. Fisch
3331	J. Chester
3331	W. Hauck
3331	P. Koenigs
3331	M. Fecher
3331	J. Monti
3331	J. Syck
3332	Dr. H. Weinberg
3332	G. Botseas
3332	Dr. D. Lee
3332	M. Goldstein (15)
3332	W. Kanabis
3332	H. Sternberg
3332	R. Deavenport
3332	Dr. D. Wood
3332	R. Drinkard
3332	J. Nordquist
3332	E. Robinson
3332	E. Jensen
3332	C. Turner
3332	L. Petitpas

3333	C. Brown
3333	G. Brown
3333	F. Farmer
3333	E. Montavon
3333	J. Prentice
3333	W. Sternberg
3333	W. Wachter
3333	L. Walker
3334	J. Hall
3334	E. Gannon
3334	V. Edwards
3334	Dr. D. Klingbeil
3334	F. McMullen
34	Dr. D. Dence
	J. Katan
	Dr. D. Fessenden
	K. Hafner
	J. Casey
	A. Bruno
	D. Dixon
35	D. Cardin
	T. Conrad
	L. Cabral
36	Dr. J. Sirmalis
	J. Griffin
	D. Blundell
	M. Lydon
	S. Wax
	S. Ashton
	R. McMahon
	S. Meyers
	K. Padolino
37	C. Curtis
	Q. Huynh
38	J. Kyle
401	A. Carlson
	J. Clark
	Dr. A. Kalinowski
	Dr. R. Kasper
	R. Manstan
	R. Munn
	Dr. J. Patel
	A. Shigematsu
	Dr. M. Tucchio
402	M. Berger
4111	S. Horvitz
4331	B. Antrim
	S. Walsh
434	K. Steele
	G. Lussier
60	Dr. J. Cohen
70	G. Bain
701	G. Elias
72	G. Daglieri
	R. Wilson
	S. Schneller
	J. Auwood
	T. Wheeler
	P. Breslin
	A. Alfiero
	J. Gribbin
73	S. Capizzano
	R. Pingree